F7.1SC3 Tutorial 3 - Solutions

Question 1

```
> f1:=cos(x)/(x-Pi/2);
```

$$f1 := \frac{\cos(x)}{x - \frac{\pi}{2}}$$

```
> limit(f1,x=Pi/2);
```

$$-1$$

```
> f2:=(1+r)^(2/3)-r^(2/3);
```

$$f2 := (1+r)^{(2/3)} - r^{(2/3)}$$

```
> limit(f2,r=infinity);
```

$$0$$

Question 2

```
> f:=(1+x^2)*sin(x);
```

$$f := (1+x^2)\sin(x)$$

```
> Diff(f,x)=diff(f,x);
```

$$\frac{d}{dx}((1+x^2)\sin(x)) = 2\,x\sin(x) + (1+x^2)\cos(x)$$

```
> Diff(f,x,x)=diff(f,x,x);
```

$$\frac{d^2}{dx^2}((1+x^2)\sin(x)) = 2\sin(x) + 4\,x\cos(x) - (1+x^2)\sin(x)$$

```
> Diff(f,x$3)=diff(f,x$3);
```

$$\frac{d^3}{dx^3}((1+x^2)\sin(x)) = 6\cos(x) - 6\,x\sin(x) - (1+x^2)\cos(x)$$

```
> simplify(%);
```

$$\frac{d^3}{dx^3}(\sin(x) + \sin(x)\,x^2) = 5\cos(x) - 6\,x\sin(x) - \cos(x)\,x^2$$

Question 3

```
> diff(x^x,x);
```

$$x^x(\ln(x)+1)$$

```
> diff(ln(t+sqrt(1+t^2)),t);
```

$$\frac{1 + \frac{t}{\sqrt{1+t^2}}}{t + \sqrt{1+t^2}}$$

```
> # looks messy, try simplifying
> simplify(%);
```

$$\frac{1}{\sqrt{1+t^2}}$$

```
> diff(cos(sin(x)),x);
```

$$-\sin(\sin(x))\cos(x)$$

Question 4

```
> f:=1/sqrt(Pi+x^2);
```

$$f := \frac{1}{\sqrt{\pi+x^2}}$$

```
> Int(f,x=0..1); int(f,x=0..1);
```

$$\int_0^1 \frac{1}{\sqrt{\pi+x^2}}\,dx$$

$$\operatorname{arcsinh}\left(\frac{1}{\sqrt{\pi}}\right)$$

```
> evalf(%,10);
```

$$0.5378762441$$

```
> Int(tanh(x),x=0..1); int(tanh(x),x=0..1);
```

$$\int_0^1 \tanh(x)\,dx$$

$$-\ln(2)+\ln(\mathbf{e}^{(-1)}+\mathbf{e})$$

```
> evalf(%,10);
```

$$0.4337808304$$

Question 5

```
> plot({x,x^2},x=-2..2);
```




```
> # area enclosed between x=0 and x=1 so we want to compute:
> int(x-x^2,x=0..1);
```

$$\frac{1}{6}$$

Question 6

```
> ode1:=diff(y(x),x)=(x^3*cos(x)+x^2*sin(x))*y(x);  # remember y(x)
  and not just y
```

$$ode1 := \frac{d}{dx}\,y(x) = (x^3 \cos(x) + x^2 \sin(x))\, y(x)$$

```
> dsolve(ode1,y(x));
```

$$y(x) = \_C1\, \mathbf{e}^{(x^3 \sin(x) + 2\,x^2 \cos(x) - 4\cos(x) - 4\,x \sin(x))}$$

Question 7

```
> # add in initial condition for the equation above
> ic1:=y(0)=1;
```

$$ic1 := y(0) = 1$$

```
> sol1:=dsolve({ode1,ic1},y(x));
```

$$sol1 := y(x) = \frac{\mathbf{e}^{(x^3 \sin(x) + 2\,x^2 \cos(x) - 4\cos(x) - 4\,x \sin(x))}}{\cosh(4) - \sinh(4)}$$

```
> evalf(subs(x=1,rhs(sol1)));
```

$$1.484358026$$

```
> evalf(subs(x=Pi,rhs(sol1)));
```

$$0.7974920560\ 10^{-5}$$

```
> plot(rhs(sol1),x=0..Pi);
```



Question 8

```
> ode2:=diff(y(x),x,x)+3*diff(y(x),x)-4*y(x)=0;
```

$$ode2 := \left(\frac{d^2}{dx^2}\,y(x)\right) + 3\left(\frac{d}{dx}\,y(x)\right) - 4\,y(x) = 0$$

```
> dsolve(ode2,y(x));
```

$$y(x) = \_C1\,\mathbf{e}^x + \_C2\,\mathbf{e}^{(-4\,x)}$$

```
> bcs2:=y(0)=0,y(1)=1;
```

$$bcs2 := y(0) = 0,\ y(1) = 1$$

```
> sol2:=dsolve({ode2,bcs2},y(x));
```

$$sol2 := y(x) = \frac{\mathbf{e}^x}{\mathbf{e} - \mathbf{e}^{(-4)}} - \frac{\mathbf{e}^{(-4\,x)}}{\mathbf{e} - \mathbf{e}^{(-4)}}$$

```
> simplify(rhs(sol2));
```

$$\frac{\mathbf{e}^4\,(\mathbf{e}^x - \mathbf{e}^{(-4\,x)})}{\mathbf{e}^5 - 1}$$

```
> plot(rhs(sol2),x=0..2);
```



Question 9

```
> # in this case y is a function of t
> ode3:=diff(y(t),t,t)+4*diff(y(t),t)+9*y(t)=exp(t);
```

$$ode3 := \left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 4\left(\frac{d}{dt}\,\mathrm{y}(t)\right) + 9\,\mathrm{y}(t) = \mathbf{e}^{t}$$

```
> ics3:=y(0)=0,D(y)(0)=1;
```

$$ics3 := \mathrm{y}(0) = 0,\ \mathrm{D}(y)(0) = 1$$

```
> sol3:=dsolve({ode3,ics3},y(t));
```

$$sol3 := \mathrm{y}(t) = \frac{11}{70}\,\mathbf{e}^{(-2\,t)}\,\sin(\sqrt{5}\,t)\,\sqrt{5} - \frac{1}{14}\,\mathbf{e}^{(-2\,t)}\,\cos(\sqrt{5}\,t) + \frac{1}{14}\,\mathbf{e}^{t}$$

```
> plot(rhs(sol3),t=0..2);
```



Question 10

```
> rr1:=a(n+1)=3*a(n)+4;
```

$$rr1 := \mathrm{a}(n+1) = 3\,\mathrm{a}(n) + 4$$

```
> rsolve(rr1,a(n));  # general solution
```

$$\mathrm{a}(0)\,3^n - 2 + 2\,3^n$$

```
> ic1:=a(0)=6;
```

$$ic1 := \mathrm{a}(0) = 6$$

```
> rsolve({rr1,ic1},a(n));  #  particular solution
```

$$8\,3^n - 2$$

Question 11

```
> rr2:=p(n+1)=p(n)+(n-1)^2;
```

$$rr2 := \mathrm{p}(n+1) = \mathrm{p}(n) + (n-1)^2$$

```
> ic2:=p(0)=3;
```

$$ic2 := \mathrm{p}(0) = 3$$

```
> rsolve(rr2,p(n));  #  general solution
```

$$\mathrm{p}(0) + 9\,n + 5 + 2\,(n+1)\left(\frac{n}{2}+1\right)\left(\frac{n}{3}+1\right) - 7\,(n+1)\left(\frac{n}{2}+1\right)$$

```
> rsolve({rr2,ic2},p(n));  # particular solution
```

$$8 + 9\,n + 2\,(n+1)\left(\frac{n}{2}+1\right)\left(\frac{n}{3}+1\right) - 7\,(n+1)\left(\frac{n}{2}+1\right)$$

```
> # looks like there might be some simplification possible
> simplify(%);
```

$$3 + \frac{13}{6}\,n + \frac{1}{3}\,n^3 - \frac{3}{2}\,n^2$$

Question 12

```
> # following the rules given we have a(n+1)=a(n)+a(n-1), and
  a(1)=a(2)=1 so:
> rr3:=a(n+1)=a(n)+a(n-1);
```

$$rr3 := a(n+1) = a(n) + a(n-1)$$

```
> ic3:=a(1)=1,a(2)=1;
```

$$ic3 := a(1) = 1,\ a(2) = 1$$

```
> fib:=rsolve({rr3,ic3},a(n));
```

$$fib := -\frac{\sqrt{5}\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^{n}}{5}+\frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{n}}{5}$$

```
> subs(n=3,fib);
```

$$-\frac{\sqrt{5}\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^{3}}{5}+\frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{3}}{5}$$

```
> simplify(%);
```

$$2$$

```
> # so the third term in the series is 2 as expected
> subs(n=25,fib);
```

$$-\frac{\sqrt{5}\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^{25}}{5}+\frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^{25}}{5}$$

```
> simplify(%);
```

$$75025$$

```
> # and the 25th term is 75025
```